

Zig: il controllo e la potenza del C, senza spararsi sui piedi 🚬

- zig-bolognajs.vercel.app 🔗
- [jackdbd/zig-bolognajs](https://github.com/jackdbd/zig-bolognajs) 🔗

Giacomo Debidda

Freelance full stack developer / web performance consultant

I write TypeScript / Clojure / Zig

I like:

- 📖 - add me on goodreads 🔗
- 🛹 - surfskating 🔗 actually
- 🛼 - it's time for a rollerblading emoji 🍻

🐙 jackdbd

🐦 jackdbd

👤 giacomodebidda.com



Why should we care about compiled languages?




1. **Performance:** Compiled languages like C++, Rust, or Go often offer better performance than interpreted languages. This can be important for **CPU-intensive tasks, algorithms, or systems programming**. ”
2. **Understanding of low-level details:** Learning a compiled language can give you a **better understanding of what's happening under the hood** in your system. This can help you write more efficient code, even in higher-level languages like JavaScript.
3. **Interoperability:** Node.js supports native addons, which are **dynamically-linked shared objects** written in C++. These addons can be used to perform tasks that are either more efficiently done in C++, or simply not possible in JavaScript.
4. **Type Safety:** Many compiled languages are statically typed, meaning type checking is done at compile time. This can help catch errors earlier in the development process. While JavaScript is dynamically typed, **understanding static typing systems** can still be beneficial, especially with the rise of TypeScript in the JavaScript ecosystem.
5. **Career Flexibility:** Having experience in both interpreted and compiled languages can make you a more versatile developer and open up **more job opportunities**.
6. **Different Paradigms:** Many compiled languages use **different programming paradigms** (like functional or procedural programming) than JavaScript. Learning these can broaden your problem-solving skills and help you write better code in any language.
7. **WebAssembly:** This is a binary instruction format that allows code written in languages like C, C++, and Rust to run in the browser alongside JavaScript. As a Node.js developer, **learning about WebAssembly** can be beneficial as it becomes more prevalent in web development.

—GPT 4





Source: Why should a Node.js developer care about a compiled language?

Why not C or C++?

C


- Footguns everywhere. See [banned.h](#)  for a list of functions banned in the git codebase
- Preprocessor macros. The C preprocessor is another language 
- Cleanup code can be really messy 
- A lot of undefined behavior

C++

- Complex, too many features
- Error handling typically done using exceptions (the Google C++ Style Guide prohibits their use)
- Why should I have written ZeroMQ in C, not C++, part 1  and part 2 
- OOP cargo cult: The religion of virtual methods , RAII (Resource Acquisition Is Initialization) 

Why not Rust or Go?


Rust

- Complex, too many features
- Ownership and lifetimes are hard to understand
- Writing unsafe Rust is hard because it has a lot of nuanced rules about undefined behaviour
- Questionable policies 

Go

- Garbage collected
- "Closed-world" language

C++, Rust, and D have such a large number of features that they can be distracting from the actual meaning of the application you are working on. One finds oneself debugging one's knowledge of the programming language instead of debugging the application itself.

Source: Why Zig When There is Already C++, D, and Rust? 

**SMALL LANGUAGE,
SIMPLE SYNTAX,
BUILD TOOLCHAIN**



**NO HIDDEN
CONTROL FLOW,
NICE ERROR HANDLING,
MEMORY ALLOCATORS**



**COMPILE-TIME
EVALUATION,
BUILT-IN
CROSS-COMPILATION**



**GREAT C-INTEROP,
FIRST-CLASS
WASM SUPPORT**



Learn Zig to learn how computers work

Learn C to learn about how computers work.


—A lot of people on the internet

Source: Reddit, Twitter, etc



Learn C to learn **more** about how computers work.

—Steve Klabnik

Source: Should you learn C to "learn how the computer works"? 



Learn Zig to learn **even more** about how computers work*.

—Me

Source: This slide



* Because you have to pick your own memory allocator, your own libc, etc.

Keywords

- | | | | |
|---------------------------|---------------------------|------------------------------|---------------------------------|
| 1. <code>addrspace</code> | 14. <code>const</code> | 27. <code>linksection</code> | 40. <code>switch</code> |
| 2. <code>align</code> | 15. <code>continue</code> | 28. <code>noalias</code> | 41. <code>test</code> |
| 3. <code>allowzero</code> | 16. <code>defer</code> | 29. <code>noinline</code> | 42. <code>threadlocal</code> |
| 4. <code>and</code> | 17. <code>else</code> | 30. <code>nosuspend</code> | 43. <code>try</code> |
| 5. <code>anyframe</code> | 18. <code>enum</code> | 31. <code>opaque</code> | 44. <code>union</code> |
| 6. <code>anytype</code> | 19. <code>errdefer</code> | 32. <code>or</code> | 45. <code>unreachable</code> |
| 7. <code>asm</code> | 20. <code>error</code> | 33. <code>orelse</code> | 46. <code>usingnamespace</code> |
| 8. <code>async</code> | 21. <code>export</code> | 34. <code>packed</code> | 47. <code>var</code> |
| 9. <code>await</code> | 22. <code>extern</code> | 35. <code>pub</code> | 48. <code>volatile</code> |
| 10. <code>break</code> | 23. <code>fn</code> | 36. <code>resume</code> | 49. <code>while</code> |
| 11. <code>callconv</code> | 24. <code>for</code> | 37. <code>return</code> | |
| 12. <code>catch</code> | 25. <code>if</code> | 38. <code>struct</code> | |
| 13. <code>comptime</code> | 26. <code>inline</code> | 39. <code>suspend</code> | |

What Zig leaves out

We need to build **simple systems** if we want to build **good systems**.

The benefits of simplicity are: ease of understanding, ease of change, ease of debugging, flexibility.

—Rich Hickey

Source: Simple Made Easy 

”

No garbage collection

Modern garbage collectors (G1, Orinoco, etc) are really complex and they are basically a black box.

When, how, and whether garbage collection occurs is down to the implementation of any given JavaScript engine. Any behavior you observe in one engine may be different in another engine, in another version of the same engine, or **even in a slightly different situation with the same version of the same engine.**

Source: [FinalizationRegistry on mdn web docs](#)

It is **not guaranteed** that `__del__()` methods are called for objects that still exist when the interpreter exits.

Source: [__del__ on docs.python.org](#)

No reference counting

In C++, `shared_ptr` uses automatic reference counting.

Several C libraries (GTK, Cairo) use `GObject` (GLib Object System). `GObjects` are reference counted. As long as their reference count is nonzero, they are "alive", and when their reference count drops to zero, they are deleted from memory.



`GObject`'s use of GLib's `g_malloc()` memory allocation function will cause the program to **exit unconditionally upon memory exhaustion.**

Many programming languages choose to handle the possibility of heap allocation failure by unconditionally crashing. By convention, Zig programmers do not consider this to be a satisfactory solution.



Source: [Heap Allocation Failure on ziglang.org](#)

Explicit allocations

Explicit memory management is hard, right? Not necessarily.

- [Code for Game Developers - Anatomy of a Memory Allocation \(Jorge Rodriguez\)](#) 
- [Introduction to General Purpose Allocation \(Casey Muratori\)](#) 




Understanding a generational garbage collector like Orinoco is much harder.

- [Orinoco: The new V8 Garbage Collector \(Peter Marshall\)](#) 
- [Garbage Collection Algorithms \(Dmitry Soshnikov\)](#) 



Allocator interface

In Zig, functions which need to allocate accept an `Allocator` parameter.

The memory allocator interface is defined in `std/mem/Allocator.zig`  and `std/mem.zig` .

- [What's a Memory Allocator Anyway? \(Benjamin Feng\)](#) 
- [Testing memory allocation failures with Zig](#) 
- [Choosing an Allocator](#) 

This is a good idea. In fact, others are taking notes:

- [fitzgen/bumpalo](#) 
- [Trait std::alloc::Allocator](#) 

std.heap

Memory allocators in `std/heap.zig`:

- `std.heap.ArenaAllocator`
- `std.heap.FixedBufferAllocator`
- `std.heap.GeneralPurposeAllocator`
- `std.heap.LoggingAllocator`
- `std.heap.LogToWriterAllocator`
- `std.heap.PageAllocator`
- `std.heap.SbrkAllocator`
- `std.heap.ScopedLoggingAllocator`
- `std.heap.ThreadSafeAllocator`
- `std.heap.WasmAllocator`
- `std.heap.WasmPageAllocator`

std.testing

Memory allocators in `std/testing.zig`:


- `std.testing.allocator`
- `std.testing.FailingAllocator`

No string type 1/2

How long is the string 北斗の拳 ?

It does not make sense to have a string without knowing what **encoding** it uses.

—Joel Spolsky

Source: The Absolute Minimum Every Software Developer Absolutely, Positively Must Know About Unicode and Character Sets (No Excuses!) 

No string type 2/2

The encoding of a string in Zig is de-facto assumed to be UTF-8. See String Literals and Unicode Code Point Literals  in the documentation.

- 北斗の拳 is 4 code points long.
- 北 is a Japanese Kanji and takes 3 bytes in UTF-8.
- 斗 and 拳 are also Japanese Kanji, so 3 bytes each.
- の is a Japanese Hiragana character and takes 3 bytes in UTF-8.

So 北斗の拳 12 u8 long in UTF-8.

Trivia: there are many Japanese encoding standards

- JIS X 0208 is a 2-byte character set.
- ISO-2022-JP uses 7 bits.

No operator overloading

What does this Python code print?

```
1 a = Foo(2)
2 b = Bar(3)
3 print(a + b)
4 print(b + a)
```

We need to know what `+` means for `a` and `b`.




```
1 class Foo(object):
2     def __init__(self, n):
3         self.n = n
4     def __add__(self, other):
5         return self.n + other.n
6
7 class Bar(object):
8     def __init__(self, n):
9         self.n = n
10    def __add__(self, other):
11        return self.n - other.n
```

Solution:

```
1 5
2 1
```

Why not?

Arguments in favor of / against operator overloading.

- Proposal: Custom Operators / Infix Functions (issue #427) 
- Operator Overloading (issue #871) 
- New to Zig. I had some questions and comments (r/Zig) 

No exceptions

Exceptions make cleaning up resources problematic.

When you add a `throw` statement to an existing function, you” must examine all of its transitive callers. Either they must make at least the basic exception safety guarantee, or they must never catch the exception and be happy with the program terminating as a result. For instance, if `f()` calls `g()` calls `h()`, and `h` throws an exception that `f` catches, `g` has to be careful or it may not clean up properly.

Source: [Google C++ Style Guide](#) @

Exceptions hide control flow.

[...] exceptions make the control flow of programs difficult to evaluate by looking at code: functions may return in places you don't expect. This causes maintainability and debugging difficulties.”

Source: [Google C++ Style Guide](#) @

Errors as values

Zig errors are **values**. Handle them like any other value.

```
1  fn doAThing(str: []u8) void {
2      if (parseU64(str, 10)) |number| {
3          doSomethingWithNumber(number);
4      } else |err| switch (err) {
5          error.Overflow => {
6              // handle overflow...
7          },
8          error.InvalidChar => {
9              // handle invalid char...
10         },
11         // Zig will not compile if we forgot to
12         // handle all possible errors.
13     }
14 }
```

We can use the same approach in our JS code by never throwing an `Error`, but always returning it.


For example, in Joi:

```
1  const result = joi.validate(value, options)
2  const { error, value, warning, artifacts } = result
```

Error handling: vs

In order to have high quality software, correct error handling has to be the **easiest, most straightforward path** for people to follow. ”


—Andrew Kelley


Source: The Road to Zig 1.0 

Defining errors in

Consider extending the `Error` object with additional properties,[”] but be careful not to overdo it. It's generally a good idea to extend the built-in `Error` object **only once**.

—nodebestpractices

Source: Use only the built-in `Error` object 

Boom (Hapi.js)  represents all HTTP errors with a **single** class that extends `Error`.

```
1 exports.Boom = class extends Error {
2   constructor(messageOrError, options = {}) {
3     // ...
4   }
5 }
```

@fastify/error  defines this factory function.

```
1 function createError (
2   code, message, statusCode = 500, Base = Error
3 ) {
4   // ...
5 }
```

Defining errors in



Define an error set type 

```
1 const NumberNotInRangeError = error{
2   TooSmall,
3   TooBig,
4 };
```

The return type of a Zig function that might fail is:

```
1 <error set type>!<expected type>
```

Zig errors cannot have a payload.

- Some people would want it 
- Some others would not 

Handling failures in

In JavaScript, `catch` catches **exceptions**.

JS functions can `throw` anything → An exception can be anything.

We **do not know** what we caught.

```
1  const fn = () => {
2    throw "I'm not an error object"
3    // throw 42
4    // throw true
5    // throw { a: 1 }
6    // throw undefined
7  }
8
9  const main = async () => {
10   try {
11     fn()
12   } catch (ex) {
13     console.trace(ex)
14     console.log("message", ex.message)
15     console.log("stack trace", ex.stack)
16   }
17 }
18
19 main()
```

Handling failures in

In Zig, `catch` catches **errors**.

Zig functions can `return` possible error values → An error type is a set of all possible error values.

We **know** what we caught.

```
1  fn isNumInRange(n: u8) NumberNotInRangeError!bool {
2    if (n ≤ 3) {
3      return NumberNotInRangeError.TooSmall;
4    } else if (n ≥ 7) {
5      return NumberNotInRangeError.TooBig;
6    } else {
7      return true;
8    }
9  }
10
11 pub fn main() void {
12   var b = isNumInRange(5) catch false;
13   std.debug.print("5 in range? {}\n", .{ b });
14
15   b = isNumInRange(9) catch |err| blk: {
16     std.debug.print("Error: {any}\n", .{err});
17     break :blk false;
18   };
19   std.debug.print("9 in range? {}\n", .{ b });
20 }
```

defer and errdefer

Allocate, then `defer` a deallocation immediately afterwards.

Cleanup resources using `errdefer`.

```
1  fn createFoo(param: i32) !Foo {
2      const foo = try tryToAllocateFoo();
3
4      // Now we have allocated foo. We need to free it if the function fails.
5      // But we want to return it if the function succeeds.
6      errdefer deallocateFoo(foo);
7
8      const tmp_buf = allocateTmpBuffer() orelse return error.OutOfMemory;
9      // tmp_buf is truly a temporary resource, and we for sure want to clean it up
10     // before this block leaves scope.
11     defer deallocateTmpBuffer(tmp_buf);
12
13     if (param > 1337) return error.InvalidParam;
14
15     // The errdefer will not run since we're returning success from the function.
16     // But the defer will run!
17     return foo;
18 }
```

This passage of The Road to Zig 1.0  explains well how `errdefer` works.


try / catch / @panic / @compileError

The keyword `try` is a shortcut for `catch |err| return err`. That `|err|` is called **capture**.



Often you don't `catch`. You simply `try`. You `catch` only **when you can handle** the error.

If you have **no idea how to handle** a runtime error and/or **want to crash** the program, use `@panic`.

You should (ideally) never use `@panic` in a library.

You can override the behavior  of `@panic`. I'm not sure it's a good idea though.

If you know already at **compile time** that something is wrong, use `@compileError`.

- Error, panic or unreachable? - Loris Cro 
- Zig / Handling errors 

error return trace \neq stack trace

When an error is returned, you get an **error return trace**.

When `@panic` is called, you get a **stack trace**.

This comparison  illustrates how an error return trace offers better debuggability.

Tips for error handling 1/2

✓ Do omit the error set of a function.


```
1 pub fn foo() !u32 {  
2     // ...  
3 }
```

Even in recursive functions.

```
1 const MyError = error{  
2     FourIsBadLuck,  
3 };  
4  
5 fn factorial(n: usize) !usize {  
6     if (n == 1) return 1;  
7     if (n == 4) return MyError.FourIsBadLuck;  
8     return n * try factorial(n - 1);  
9 }
```

✗ Do not use `anyerror` as the error set.

```
1 pub fn foo() anyerror!u32 {  
2     // ...  
3 }
```



The global error set `anyerror` should generally be avoided  because it prevents the compiler from knowing what errors are possible at compile-time.

Knowing the error set at compile-time is better for generated documentation and helpful error messages.

Build a JS project

`package.json` is a **manifest**.

You build your project using something else:


- npm scripts
- bash scripts
- make
- webpack
- esbuild
- pkg 
- Single Executable Applications 
- etc

Build a Zig project

`build.zig` is a **program**.

Zig is not the only language that takes this approach.

The philosophy behind tools.build is that your project build is inherently a program - a series of instructions to create one or more project artifacts from your project source files. ”

Source: Builds are programs 

You build your project using the Zig compiler and toolchain. Building should be as simple as running:

```
1  zig build
2  # which stands for:
3  zig build --build-file build.zig
```

No need for extra build tools (make, Ninja) or meta-build tools (CMake, Meson, gn).

Let's use  in  with 


WebAssembly

Zig supports building for WebAssembly out of the box 

Browsers

```
1  zig build-lib src/lib.zig \  
2    -target wasm32-freestanding -dynamic \  
3    -O ReleaseSmall \  
4    --export format_zig_code \  
5    --export wasm_alloc \  
6    --export wasm_dealloc
```

Generates `lib.wasm`.


WASI runtimes (WASI support is under active development )

```
1  zig build-exe src/main.zig \  
2    -target wasm32-wasi-musl \  
3    -O ReleaseFast
```

Generates `main.wasm`.

Zig formatter in WebAssembly

Source code: [jackdbd/zigfmt-web](https://github.com/jackdbd/zigfmt-web) 

Paste some unformatted zig code in the `textarea` below and click  to format it using `fmt.wasm`.


```
const std = @import("std"); pub fn main() void { std.debug.warn("Hello World\n");
}
```

Logs...



Debug it! Open Chrome DevTools > `Sources` tab > open `fmt.wasm` > place a breakpoint in the `$format_zig_code` function.

Checkerboard



Debug it!

- Open Chrome DevTools.
- Go to the `Sources` tab.
- Open the `checkerboard.wasm` file.
- Place a breakpoint in the `$colorCheckerboard` function, right after the local variables.
- Press F8 a few times to resume script execution when it pauses on the breakpoint.

Checkerboard (JS)

Instantiate a WebAssembly module and grant it an initial and maximum number of WebAssembly.Memory [pages](#) (64 KiB each).

```
1 let memory = new WebAssembly.Memory({
2   initial: 2, maximum: 2
3 })
4
5 const importObject = { env: { memory } }
6
7 const {
8   instance
9 } = await WebAssembly.instantiateStreaming(
10  fetch('checkerboard.wasm'), importObject)
```

These functions are defined in `checkerboard.zig` using `export fn`.

```
1 const {
2   colorCheckerboard,
3   getCheckerboardBufferPointer,
4   getCheckerboardSize,
5 } = instance.exports
```

Get the canvas from the DOM, get a memory slice representing a 1D RGBA array, use this slice as image data, draw the canvas.

Checkerboard (Zig)

Define functions and `export` them.

```
1 const std = @import("std");
2
3 const checkerboard_size: usize = 8;
4
5 // each pixel is 4 bytes (rgba)
6 var checkerboard_buffer = std.mem.zeroes(
7   [checkerboard_size][checkerboard_size][4]u8,
8 );
9
10 // The returned pointer will be used as an offset
11 // integer to the wasm memory
12 export fn getCheckerboardBufferPointer() [*]u8 {
13   return @ptrCast(&checkerboard_buffer);
14 }
15
16 export fn getCheckerboardSize() usize {
17   return checkerboard_size;
18 }
19
20 export fn colorCheckerboard(
21   dark_value_red: u8,
22   dark_value_green: u8,
23   dark_value_blue: u8,
24   light_value_red: u8,
25   light_value_green: u8,
26   light_value_blue: u8,
27 ) void {
28   // implementation not shown
29 }
```

Checkerboard (wasm)

We can inspect the sections available in the WASM binary using the WebAssembly Binary Toolkit .

The `Export` section contains zig functions we exported using `export fn`.

```
1  wasm-objdump public/checkerboard.wasm --details --section Export
2
3  Export[3]:
4    - func[0] <getCheckerboardBufferPointer> → "getCheckerboardBufferPointer"
5    - func[1] <getCheckerboardSize> → "getCheckerboardSize"
6    - func[2] <colorCheckerboard> → "colorCheckerboard"
```



The `Import` section contains functions and `WebAssembly.Memory` objects we defined in JS.

```
1  wasm-objdump public/checkerboard.wasm --details --section Import
2
3  Import[1]:
4    - memory[0] pages: initial=2 max=2 ← env.memory
```

The `Type` section contains the data types available in the wasm module.

```
1  wasm-objdump public/checkerboard.wasm --details
2
3  Type[2]:
4    - type[0] () → i32
5    - type[1] (i32, i32, i32, i32, i32, i32) → nil
6  Function[3]:
7    - func[0] sig=0 <getCheckerboardBufferPointer>
8    - func[1] sig=0 <getCheckerboardSize>
9    - func[2] sig=1 <colorCheckerboard>
```

wasm-api

Render DOM elements in WASM using wasm-api . Kind of how Yew (Rust)  works.

Click to

create a canvas

and

remove it from the DOM

wasm-api (JS/TS)

```
1 import {WasmBridge} from "@thi.ng/wasm-api"
2 import {WasmDom} from "@thi.ng/wasm-api-dom"
3 import {WasmCanvas2D} from "@thi.ng/wasm-api-canvas"
4 import WASM_URL from "/canvas.wasm?url"
5
6 const bridge = new WasmBridge([
7   new WasmCanvas2D(), new WasmDom()
8 ])
9
10 await bridge.instantiate(fetch(WASM_URL))
11
12 // call WASM main function to kick off
13 bridge.exports.start()
```

wasm-api (Zig)

```
1 const canvas2d = @import("wasm-api-canvas");
2 const wasm = @import("wasm-api");
3 const dom = @import("wasm-api-dom");
4
5 // expose thi.ng/wasm-api core API
6 // (incl. panic handler & allocation fns)
7 pub usingnamespace wasm;
8
9 fn initApp() !void {
10     const canvas = dom.createCanvas(&.{
11         .width = 640,
12         .height = 480,
13         .parent = dom.body,
14         .dpr = 1,
15         .index = 0,
16     });
17
18     canvas2d.beginCtx(canvas);
19
20     canvas2d.setFont("100px Menlo");
21     canvas2d.setTextBaseline(.top);
22     canvas2d.fillText("Ciao", 10, 10, 0);
23 }
24
25 export fn start() void {
26     wasm.printStr("started canvas-wasm");
27     initApp() catch |e| @panic(@errorName(e));
28 }
```

Let's use  in 

Calling Zig from Node.js


Possible approaches:

1. Direct use of internal V8, libuv, and Node.js libraries
2. Native Abstractions for Node.js (nan)
3. Node-API (formerly known as N-API)

Node-API versions are additive and versioned independently from Node.js. See [Node-API version matrix](#) 

Version 9 is an extension to version 8 in that it has all of the APIs from version 8 with some additions.

Unless there is a need for direct access to functionality which is not exposed by Node-API, use **Node-API**. 

Source: [Node.js documentation](#) 

zig translate-c 1/4

Let's say we have downloaded the Node.js 18.17.0 header files to `deps/node-v18.17.0`.

```
1  .
2  └─ include
3      └─ node
4          ├── common.gypi
5          ├── config.gypi
6          ├── cppgc
7          ├── js_native_api.h
8          ├── js_native_api_types.h
9          ├── ...
10         ├── node_api.h
11         ├── node_api_types.h
12         ├── ...
13         ├── uv.h
14         ├── v8.h
15         ├── ...
16         └─ zlib.h
```

We can try translating the header files from C to Zig:

```
1  zig translate-c \
2      deps/node-v18.17.0/include/node/node_api.h \
3      > node_api.zig
```

zig translate-c 2/4

The C enum `napi_status`...

```
1  typedef enum {
2      napi_ok,
3      napi_invalid_arg,
4      napi_object_expected,
5      napi_string_expected,
6      // etc ...
7  } napi_status;
```

...becomes a bunch of Zig constants:

```
1  pub const napi_ok: c_int = 0;
2  pub const napi_invalid_arg: c_int = 1;
3  pub const napi_object_expected: c_int = 2;
4  pub const napi_string_expected: c_int = 3;
5  // etc ...
6  pub const napi_status = c_uint;
```


zig translate-c 3/4

The C function napi_create_string_utf8...

```
1  NAPI_EXTERN napi_status NAPI_CDECL
2  napi_create_string_utf8(
3      napi_env env,
4      const char* str,
5      size_t length,
6      napi_value* result
7  );
```

...becomes this Zig function:


```
1  pub extern fn napi_create_string_utf8(
2      env: napi_env,
3      str: [*c]const u8,
4      length: usize,
5      result: [*c]napi_value
6  ) napi_status;
```

zig translate-c 4/4

The translation might not be perfect, so check:

```
1  cat napi_api.zig | grep 'unable to translate'
```

We can use `zig translate-c` to:

- understand weird C code  (e.g. learn about the symbols exported by a C library)
- produce Zig code before editing it into more idiomatic code (e.g. when we want to create a Zig wrapper for a C library)

For example, we could convert napi_status into this:

```
1  pub const Status = enum(u5) {
2      ok = c.napi_ok,
3      invalid_arg = c.napi_invalid_arg,
4      object_expected = c.napi_object_expected,
5      string_expected = c.napi_string_expected,
6      // etc
7  };
```

Create a Zig **wrapper** of a C library

Why?

- Namespaces
- Better error handling
- Easier and safer to use (e.g. less risk of buffer overflow)
- Extend the original library with higher level abstractions (e.g. slices instead of pointers)

How?

- How I built zig-sqlite 
- Wrapping a C Library with Zig 
- Iterative Replacement of C with Zig 


Examples:

- zig-v8 
- zig-cairo 
- zig-sqlite 
- duckdb.zig 
- base64-simd 

Node.js addon with Node-API 1/2

Possible project structure:

```
1  .
2  |— build.zig
3  |— deps
4  |   └─ node-v18.17.0
5  |— dist
6  |   └─ debug
7  |   └─ release
8  |— download-node-headers.sh
9  |— examples
10 |   └─ bare-minimum.cjs
11 |   └─ greet.cjs
12 |— package.json
13 |— package-lock.json
14 |— README.md
15 |— src
16 |   └─ addon.zig
17 |   └─ c.zig
18 |   └─ napi.zig
19 |— test
20 |   └─ index.cjs
21 |— zig-cache
22 |— zig-out
```

Source: [jackdbd/zig-nodeapi-example](https://github.com/jackdbd/zig-nodeapi-example) 

Node.js addon with Node-API 2/2

In `c.zig`, import the C header file/s.

```
1 pub usingnamespace @cImport({
2     @cInclude("node_api.h");
3 });
```


In `napi.zig`, import `c.zig` and implement all Zig data types / functions you want.

```
1 pub const c = @import("c.zig");
2
3 // implement Zig wrappers in this file.
```

In `addon.zig`, implement your addon functions and register them.


```
1 const std = @import("std");
2 const c = @import("c.zig");
3 const napi = @import("napi.zig");
4
5 fn foo(env: c.napi_env, cbinfo: c.napi_callback_info) callconv(.C) c.napi_value {
6     _ = cbinfo;
7     return napi.create_string(env, "Hi from the native addon!") catch return null;
8 }
9
10 export fn napi_register_module_v1(env: c.napi_env, exports: c.napi_value) c.napi_value {
11     napi.register_function(env, exports, "foo", foo) catch return null;
12     return exports;
13 }
```

Compilation targets



Build modes (optimizations)

Mode	Compilation speed	Safety checks	Runtime performance	Binary size	Reproducible build
Debug (default)	fast	✓	slow	large	✗
ReleaseFast	slow	✗	fast	large	✓
ReleaseSafe	slow	✓	medium	large	✓
ReleaseSmall	slow	✗	medium	small	✓


You can also use `@setRuntimeSafety(false)` to disable runtime safety checks  for individual scopes.

```
1  fn foo() void {
2      var x: u8 = 255;
3      x += 1; // undefined behavior
4      {
5          // runtime safety checks enabled, even for ReleaseFast and ReleaseSmall
6          @setRuntimeSafety(true);
7          var x: u8 = 255;
8          x += 1;
9      }
10 }
```

Let's use  in  with 

WASI (JS)

```
1 import { readFile } from "node:fs/promises"
2 import { WASI } from "node:wasi"
3
4 // argv[0] - node binary
5 // argv[1] - fullpath to this script
6 // argv[2] - first parameter to pass to the WASI program
7 const args = process.argv.slice(1);
8
9 const wasi = new WASI({
10   version: "preview1",
11   args,
12   env: process.env,
13   preopens: {
14     "/docs": "/home/jack/repos/zig-bolognajs/assets"
15   },
16 })
17
18 const wasm = await WebAssembly.compile(await readFile(new URL("main.wasm", import.meta.url)))
19
20 const instance = await WebAssembly.instantiate(wasm, wasi.getImportObject())
21
22 wasi.start(instance)
```

The WebAssembly System Interface (WASI)  is still experimental.

There is a discussion on marking the Node.js WASI module as stable .

Launch with `node --experimental-wasi-unstable-preview1 app.js` in Node.js < 20.0.0.

WASI (Zig) 1/2

Import the `wasi` module from the standard library.

```
1  const std = @import("std");
2  const wasi = std.fs.wasi;
3
4  pub fn main() !void {}
```

Implement the `main` function (1/2).

```
1  var gpa = std.heap.GeneralPurposeAllocator({});
2  defer _ = gpa.deinit();
3  var allocator = gpa.allocator();
4
5  const args = try std.process.argsAlloc(allocator);
6  defer std.process.argsFree(allocator, args);
7
8  const stdout = std.io.getStdOut().writer();
9  const stderr = std.io.getStdErr().writer();
10 var preopens = try wasi.preopensAlloc(allocator);
11
12 const fd = preopens.find("/docs");
13 if (fd == null) {
14     try std.fmt.format(
15         stderr,
16         "/docs is not an available preopen: {s}\n",
17         .{preopens.names}
18     );
19     std.process.exit(1);
20 }
```

WASI (Zig) 2/2

Implement the `main` function (2/2).

```
1  var dir = std.fs.Dir{ .fd = fd orelse unreachable };
2  defer dir.close();
3
4  var file = try dir.openFile(input_filename, .{});
5  defer file.close();
6
7  const contents = try file.reader().readAllAlloc(
8      allocator, std.math.maxInt(usize)
9  );
10 defer allocator.free(contents);
11
12 std.log.debug(
13     "Contents of {s}\n{s}",
14     .{ input_filename, contents }
15 );
```

Compile with:

```
1  zig build-exe main.zig -target wasm32-wasi
```

Every WASI-compliant runtime implements the file system interface with a `libpreopen`-like layer.

This group proposes an alternative: Leveraging eBPF to enhance sandboxing of WebAssembly runtimes (PDF).

Let's use  in other runtimes

JS+WASM runtimes

We can deploy to any environment that offers both a JS runtime **and** a WebAssembly runtime.

- AWS Lambda and AWS Lambda@Edge (V8)
- Bun (JavaScriptCore)
- Cloudflare Workers and Pages Functions (V8)
- Deno and Deno Deploy (V8)
- Fastly Compute@Edge (SpiderMonkey)
- WasmEdge (QuickJS)
- Lagon?

No need to change Zig code. Keep compiling it with `-target wasm32-wasi`.

These ones are **just** WASM runtimes:

- wasmtime
- WebAssembly Micro Runtime (WAMR)
- GraalWasm

WASI in Workers

```
1  import { WASI } from "@cloudflare/workers-wasi"
2  import demoWasm from "./demo.wasm"
3
4  export default {
5    async fetch(request, _env, ctx) {
6      const stdout = new TransformStream()
7
8      const wasi = new WASI({
9        args: [],
10        // Cloudflare does not expose a filesystem API
11        // ⇒ no WASI preopens except stdin/stdout
12        // See Cloudflare security-model
13        stdin: request.body,
14        stdout: stdout.writable,
15      })
16
17      const instance = new WebAssembly.Instance(
18        demoWasm,
19        { wasi_snapshot_preview1: wasi.wasiImport }
20      )
21
22      // Keep our worker alive until the WASM has
23      // finished executing.
24      ctx.waitUntil(wasi.start(instance))
25
26      // Finally, let's reply with the WASM's output.
27      return new Response(stdout.readable)
28    },
29  }
```

How to get zig?

Download and manage zig compilers with zigup 

Installation

```
1 wget https://github.com/marler8997/zigup/releases/download/v2023_07_27/zigup.ubuntu-latest-x86_64.zip
2 unzip zigup.ubuntu-latest-x86_64.zip
3 chmod u+x zigup
4 mv zigup ~/bin/zigup
```

Usage

```
1 zigup fetch master
2 zigup fetch 0.11.0
3
4 zigup list
5
6 zigup default 0.12.0-dev.881+42998e637
7 zigup default 0.11.0
```

Double-check with `zig version`.

How to setup VS Code for Zig?

Install the VS Code extension [ziglang.vscode-zig](#) and declare it in your `.vscode/extensions.json`

```
1  {  
2    "recommendations": ["ziglang.vscode-zig"]  
3  }
```

`ziglang.vscode-zig` automatically installs the [Zig Language Server \(zls\)](#) for autocompletion, goto definition, formatting, etc.

How to use libraries?

Past solutions

- [nektro/zigmod](#) 
- [mattnite/gyro](#) 
- [marler8997/zig-build-repos](#) 



Evergreen solutions

- vendorization (i.e. copy external dependencies into your project itself)
- git subtree (basically vendorizing)
- git submodule



Package manager

Current status

Declare dependencies in a `build.zig.zon` file.





- [Zig Package Manager -- WTF is Zon](#) 
- [build system terminology update: package, project, module, dependency #14307](#) 

The Future










- [Run build.zig logic in a WebAssembly sandbox #14286](#) 
- [Package Manager GitHub project board](#) 

How to learn Zig?

Learn the basics


1. Familiarize yourself with the syntax: ziglearn 
2. Fix tiny broken programs: ziglings/exercises 
3. Review the main features of the language 
4. Read a few functons of the standard library 

Get better

- Watch Reading Zig's Standard Library 
- Write tests, especially allocation failures usin std.testing.FailingAllocator 
- Have a look at jackdbd/zig-demos 
- Review Type/pointer cheatsheet 
- Join r/Zig  Zigg  and/or other communities  and read/ask/answer questions
- Subscribe to Zig SHOWTIME  and Zig Meetups 

The zen of Zig

1. Communicate intent precisely.
2. Edge cases matter.
3. Favor reading code over writing code.
4. Only one obvious way to do things.
5. Runtime crashes are better than bugs.
6. Compile errors are better than runtime crashes.
7. Incremental improvements.
8. Avoid local maximums.
9. Reduce the amount one must remember.
10. Focus on code rather than style.
11. Resource allocation may fail; resource deallocation must succeed.
12. Memory is a resource.
13. Together we serve the users.

Source: Zen 



while 1/4

```
1  const std = @import("std");
2
3  pub fn main() void {
4      std.log.info("while loop [0, 7)", .{});
5      var i: usize = 0;
6      while (i < 7) {
7          std.log.debug(
8              "i ({s}): {d}",
9              .{ @typeName(@TypeOf(i)), i }
10         );
11         i += 1;
12     }
13 }
```

Output:

```
1  info: for loop [0, 7)
2  debug: i (usize): 0
3  ...
4  debug: i (usize): 7
```

while 2/4

```
1  const std = @import("std");
2
3  pub fn main() void {
4      std.log.info("while loop [0, 7)", .{});
5      var i: usize = 0;
6      while (i < 7) : (i += 1) {
7          std.log.debug(
8              "i ({s}): {d}",
9              .{ @typeName(@TypeOf(i)), i }
10         );
11     }
12 }
13 }
```

Output:

```
1  info: for loop [0, 7)
2  debug: i (usize): 0
3  ...
4  debug: i (usize): 7
```

while 3/4

```
1  const std = @import("std");
2
3  var numbers_left: u32 = undefined;
4  fn eventuallyNull() ?u32 {
5      return if (numbers_left == 0) null else blk: {
6          numbers_left -= 1;
7          break :blk numbers_left;
8      };
9  }
10
11 pub fn main() void {
12     var tot: u32 = 0;
13     numbers_left = 3;
14     while (eventuallyNull()) |value| {
15         std.log.debug("num left: {d}", .{value});
16         tot += value;
17     }
18     std.log.debug("total: {d}", .{tot});
19 }
```

Output:

```
1  debug: num left: 2
2  debug: num left: 1
3  debug: num left: 0
4  debug: total: 3
```

while 4/4

```
1  const std = @import("std");
2
3  fn inRange(i0: usize, i1: usize, n: usize) bool {
4      var i = begin;
5      return while (i < end) : (i += 1) {
6          if (i == n) {
7              break true;
8          }
9      } else false;
10 }
11
12 pub fn main() void {
13     std.log.debug(
14         "is 3 within [1, 5) ? {}",
15         .{inRange(1, 5, 3)}
16     );
17     std.log.debug(
18         "is 7 within [1, 5) ? {}",
19         .{inRange(1, 5, 7)}
20     );
21 }
```

Output:

```
1  debug: is 3 within [1, 5) ? true
2  debug: is 7 within [1, 5) ? false
```

Capture value and index

```
1  const std = @import("std");
2  const log = std.log;
3
4  pub fn main() void {
5      const colors = [_][]const u8{ "red", "green" };
6
7      for (colors, 0..) |color, i| {
8          log.debug(
9              "colors[{}d] is {}s",
10             .{ i, color }
11         );
12     }
13 }
```

Output:

```
1  debug: colors[0] is red
2  debug: colors[1] is green
```

inline for

```
1  const std = @import("std");
2  const log = std.log;
3
4  fn typeNameLength(comptime T: type) usize {
5      return @typeName(T).len;
6  }
7
8  pub fn main() void {
9      const nums = [_]i32{ 2, 4, 6 };
10
11      var sum: usize = 0;
12      inline for (nums) |n| {
13          const T = switch (n) {
14              2 => f32, // length 3
15              4 => i8,  // length 2
16              6 => bool, // length 4
17              else => @panic("got unexpected n"),
18          };
19          sum += typeNameLength(T);
20      }
21      log.debug("sum is {}d", .{sum});
22  }
```

Output:

```
1  debug: sum is 9
```

Iterate over a slice

```
1  const std = @import("std");
2  const log = std.log;
3
4  pub fn main() void {
5      var items = [_]i32{ -1, 0, 1 };
6
7      log.debug("items is {any}", .{items});
8
9      for (&items) |*val| {
10         log.debug(
11             "val has type {s} and is {}",
12             .{ @typeName(@TypeOf(val)), val }
13         );
14
15         log.debug(
16             "val.* has type {s} and is {}",
17             .{ @typeName(@TypeOf(val.*)), val.* }
18         );
19
20         // dereference and assign
21         val.* += 1;
22     }
23
24     log.debug("items is {any}", .{items});
25 }
```

Output

```
1  debug: items is { -1, 0, 1 }
2
3  debug: val has type *i32 and is i32@7ffe311feadc
4  debug: val.* has type i32 and is -1
5
6  debug: val has type *i32 and is i32@7ffe311feae0
7  debug: val.* has type i32 and is 0
8
9  debug: val has type *i32 and is i32@7ffe311feae4
10 debug: val.* has type i32 and is 1
11
12 debug: items is { 0, 1, 2 }
```

comptime

A Lisp programmer knows the value of everything, but the cost of nothing.

—Alan Perlis

Source: Epigrams on Programming



Generics

Compile-time parameters is how Zig implements generics. It is compile-time duck typing.

```
1  fn max(comptime T: type, a: T, b: T) T {
2      return if (a > b) a else b;
3  }
4
5  fn gimmeTheBiggerFloat(a: f32, b: f32) f32 {
6      return max(f32, a, b);
7  }
8
9  fn gimmeTheBiggerInteger(a: u64, b: u64) u64 {
10     return max(u64, a, b);
11 }
```

Compile-time defined types

```
1  const std = @import("std");
2
3  pub fn main() void {
4      var i: u3 = 0; // 111 in binary is 7 in decimal
5      std.log.info("while loop [0, 7)", .{});
6      while (i < 7) {
7          std.log.debug("i ({s}): {d}", .{ @typeName(@TypeOf(i)), i });
8          i += 1;
9      }
10 }
```

`u3` is not a primitive type . It's defined at compile-time by this function in std/meta.zig 

```
1  pub fn Int(comptime signedness: std.builtin.Signedness, comptime bit_count: u16) type {
2      return @Type(.{
3          .Int = .{
4              .signedness = signedness,
5              .bits = bit_count,
6          },
7      });
8  }
```

Compile-time type reflection

```
1 // demo-reflection.zig
2 const std = @import("std");
3
4 const Hello = struct {
5     foo: u32,
6     bar: []const u8,
7 };
8
9 pub fn main() void {
10     printInfoAboutStruct(Hello);
11 }
12
13 fn printInfoAboutStruct(comptime T: type) void {
14     const info = @typeInfo(T);
15     inline for (info.Struct.fields, 0..) |field, i| {
16         std.debug.print(
17             "type {s} field {d} is called '{s}' and is of type {any}\n",
18             .{ @typeName(T), i, field.name, field.type },
19         );
20     }
21 }
```

Compile and run it with `zig run demo-reflection.zig`

```
1 type demo-reflection.Hello field 0 is called 'foo' and is of type u32
2 type demo-reflection.Hello field 1 is called 'bar' and is of type []const u8
```

Read about `@typeInfo` in the documentation [📖](#).

The end

zig-bolognajs.vercel.app 